



TECHNICAL REPORT RD-ST-97-2

TACAWS/FMTI LAUNCHER SIMULATION

**David A. Bittle
Structures Directorate
Research, Development, and Engineering Center**

April 1997



U.S. ARMY MISSILE COMMAND

Redstone Arsenal, Alabama 35898-5000

Approved for public release; distribution is unlimited.

DTIC QUALITY INSPECTED 4

19970502 103

DESTRUCTION NOTICE

FOR CLASSIFIED DOCUMENTS, FOLLOW THE PROCEDURES IN DoD 5200.22-M, INDUSTRIAL SECURITY MANUAL, SECTION II-19 OR DoD 5200.1-R, INFORMATION SECURITY PROGRAM REGULATION, CHAPTER IX. FOR UNCLASSIFIED, LIMITED DOCUMENTS, DESTROY BY ANY METHOD THAT WILL PREVENT DISCLOSURE OF CONTENTS OR RECONSTRUCTION OF THE DOCUMENT.

DISCLAIMER

THE FINDINGS IN THIS REPORT ARE NOT TO BE CONSTRUED AS AN OFFICIAL DEPARTMENT OF THE ARMY POSITION UNLESS SO DESIGNATED BY OTHER AUTHORIZED DOCUMENTS.

TRADE NAMES

USE OF TRADE NAMES OR MANUFACTURERS IN THIS REPORT DOES NOT CONSTITUTE AN OFFICIAL ENDORSEMENT OR APPROVAL OF THE USE OF SUCH COMMERCIAL HARDWARE OR SOFTWARE.

UNCLASSIFIED

REPORT DOCUMENTATION PAGE				Form Approved OMB No. 0704-0188	
1a. REPORT SECURITY CLASSIFICATION UNCLASSIFIED			1b. RESTRICTIVE MARKINGS		
2a. SECURITY CLASSIFICATION AUTHORITY			3. DISTRIBUTION/AVAILABILITY OF REPORT Approved for public release; distribution is unlimited.		
2b. DECLASSIFICATION/DOWNGRADING SCHEDULE					
4. PERFORMING ORGANIZATION REPORT NUMBER(S) TR-RD-ST-97-2			5. MONITORING ORGANIZATION REPORT NUMBER(S)		
6a. NAME OF PERFORMING ORGANIZATION Structures Directorate RDEC		6b. OFFICE SYMBOL (If Applicable) AMSMI-RD-ST-SA	7a. NAME OF MONITORING ORGANIZATION		
6c. ADDRESS (City, State, and ZIP Code) Commander, U. S. Army Missile Command ATTN: AMSMI-RD-ST-SA Redstone Arsenal, AL 35898-5247			7b. ADDRESS (City, State, and ZIP Code)		
8a. NAME OF FUNDING/SPONSORING ORGANIZATION		8b. OFFICE SYMBOL (If Applicable)	9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER		
8c. ADDRESS (City, State, and ZIP Code)			10. SOURCE OF FUNDING NUMBERS		WORK UNIT ACCESSION NO.
			PROGRAM ELEMENT NO.	PROJECT NO.	
11. TITLE (Include Security Classification) TACAWS/FMTI Launcher Simulation					
12. PERSONAL AUTHOR(S) David A. Bittle					
13a. TYPE OF REPORT Final		13b. TIME COVERED FROM 05/95 TO 09/95		14. DATE OF REPORT (Year, Month, Day) April 1997	
15. PAGE COUNT 28					
16. SUPPLEMENTARY NOTATION					
17. COSATI CODES			18. SUBJECT TERMS (Continue on reverse if necessary and identify by block number)		
FIELD	GROUP	SUB-GROUP	Launcher Simulation, Future Missile Technology (FMT), Dynamics, Tipoff, The Army Combined Arms Weapon System (TACAWS)		
19. ABSTRACT (Continue on reverse if necessary and identify by block number)					
<p>This report documents the development of a dynamic model of The Army Combined Arms Weapon System (TACAWS) or Future Missile Technology Integration (FMTI) launcher. The model was developed to characterize the launch velocity and tipoff rate of the TACAWS missile and was verified by a slug missile test.</p>					
20. DISTRIBUTION/AVAILABILITY OF ABSTRACT <input checked="" type="checkbox"/> UNCLASSIFIED/UNLIMITED <input type="checkbox"/> SAME AS RPT. <input type="checkbox"/> DTIC USERS			21. ABSTRACT SECURITY CLASSIFICATION UNCLASSIFIED		
22a. NAME OF RESPONSIBLE INDIVIDUAL David A. Bittle			22b. TELEPHONE (Include Area Code) (205) 876-9410		22c. OFFICE SYMBOL AMSMI-RD-ST-SA

Previous editions are obsolete.

SECURITY CLASSIFICATION OF THIS PAGE

DD Form 1473, JUN 86

UNCLASSIFIED

i/(ii Blank)

TABLE OF CONTENTS

	<u>Page</u>
I. INTRODUCTION	1
II. DEVELOPMENT OF THE EQUATIONS OF MOTION	3
III. SIMULATION RESULTS FOR MISSILE SLUG TEST	11
IV. SIMULATION RESULTS FOR TECHNOLOGY DEMONSTRATION MISSILE	13
V. SLUG TEST TIPOFF RESULTS	15
VI. CONCLUSIONS	15
APPENDIX: SIMULATION INPUT PARAMETERS FOR THE TACAWS MISSILE SLUG	A-1

LIST OF ILLUSTRATIONS

<u>Figure</u>	<u>Title</u>	<u>Page</u>
1.	TACAWS Missile and Launch Tube.....	1
2.	Thrust Curve for TACAWS Launch Simulation.....	2
3.	TACAWS Simulation Coordinate Systems	3
4.	Detent Phase Free Body Diagram	4
5.	Tube Guidance Phase Free Body Diagram	5
6.	Tipoff Phase Kinematic Relationships	6
7.	Tipoff Phase Free Body Diagram	7
8.	Tipoff Rates for TACAWS Slug Missile - Zero QE.....	11
9.	Tipoff Rates for TACAWS Slug Missile - 10 Degree QE	12
10.	Tipoff Rate for TACAWS Technology Demonstration Missile - Zero QE	13
11.	Tipoff Rate for TACAWS Technology Demonstration Missile - 10 Degree QE.....	14

I. INTRODUCTION

This report documents the development of a dynamic model of The Army Combined Arms Weapon System (TACAWS) launcher. The model was developed in order to characterize the launch velocity and tipoff rate for the Engineering and Manufacturing Development (EMD) version of the missile, and to design a slug missile to duplicate the EMD launch.

The TACAWS system consists of an integrated missile and launch canister as shown in Figure 1, which will be fired from the TOW Missile Launcher (TML) on a Bradley Fighting Vehicle (BFV).




Figure 1. TACAWS Missile and Launch Tube

The BFV is a tracked vehicle which weighs more than 25 tons, so its movement during firing is negligible. The TML can elevate and rotate in azimuth with respect to the vehicle, but during firing it is locked rigidly in position. Upon ignition of the TACAWS launch motor, detent shear allows the missile to translate forward out of the launch canister. No missile shoes or sabots are used in this system, so the missile is free to slide out of its canister and rotate (tip) about its Center of Mass (CG).

The TACAWS system is modeled using Newtonian rigid body dynamics. The BFV and TML are considered fixed to the earth, while the TACAWS missile is allowed to translate and to rotate about a horizontal axis. The tipping of the missile over the edge of its launch canister is modeled as sliding contact. Friction between the missile and launch tube is included in the model. The TACAWS thrust curve is modeled as a function which ramps linearly up to a constant value and then shuts off. The thrust curve is illustrated in Figure 2.




Figure 2. Thrust Curve for TACAWS Launch Simulation

Input parameters for the thrust profile are t_1 , the time that constant thrust is achieved; T_{level} , the constant thrust value; and t_2 , the thrust cutoff time.

II. DEVELOPMENT OF THE EQUATIONS OF MOTION

The equations of motion for the TACAWS launch are developed with reference to the modeling diagram shown in Figure 3. Since the vehicle and launcher are considered fixed, the inertial reference origin is placed at the rear of the missile and oriented along its longitudinal axis. The reference frame for the missile is placed at the missile CG and oriented similarly.




Figure 3. TACAWS Simulation Coordinate Systems

The first phase of launch is thrust buildup prior to detent release. A free body diagram showing the forces acting at this shown in Figure 4.




Figure 4. Detent Phase Free Body Diagram

where $d2$ is the distance from the aft end of the missile to the missile CG,
 $d4$ is the length of the missile,
 F_n is the normal force,
 δ is the misalignment angle of the thrust,
 μ is the static coefficient of friction.

Since the launcher is considered fixed, the detent launch phase can be modeled statically. The time required for thrust buildup to overcome the detent is found in closed form by summation of forces.

$$\begin{aligned}\sum F_x &= \text{Thrust} \cos \delta - \mu_s F_n - W \sin(QE) - \text{Detent} = 0 \\ \sum F_y &= F_n - W \cos(QE) = 0 \\ \text{Thrust} &= [\text{Detent} + W (\mu_s \cos(QE) + \sin(QE))]/\cos \delta.\end{aligned}\tag{1}$$

The second phase of TACAWS launch is tube guidance, when the missile is supported by the launch canister. The free body diagram for this case is shown in Figure 5. Note that it is similar to the detent phase free body diagram, except that the detent force has been removed. During this launch phase the system has One Degree-of-Freedom (1-DOF).




Figure 5. Tube Guidance Phase Free Body Diagram

In Figure 5, d_1 is the distance from the bottom of the missile to the CG, d_3 is the distance from the CG to the thrust point of application, and u_1 is the vector from the CG to the point of application of the normal force.

The acceleration of the missile through the launch canister can be obtained by the following force summations:

$$\begin{aligned}\sum F_x &= \text{Thrust} \cos \delta - F_n \mu_d - W \sin(QE) = m\ddot{x} \\ \sum F_y &= F_n + \text{Thrust} \sin \delta - W \cos(QE) = 0 \\ \ddot{x} &= [\text{Thrust}(\cos \delta + \mu_d \sin \delta) - W(\mu_d \cos(QE) + \sin(QE))]/m.\end{aligned}\tag{2}$$

The velocity and displacement of the missile is obtained by numerical integration of Equation (2).

Tube guidance phase ends when the missile begins to tip relative to the launch tube. This occurs before the missile CG leaves the launch tube, since friction between the missile and launch tube causes a tipping moment. The moment balance equation, referring to Figure 5, is as follows:

$$\begin{aligned}F_n u_1 &= \mu_d F_n d_1 + \text{Thrust} (d_3 \cos \delta + d_2 \sin \delta) \\ u_1 &= [\mu_d F_n d_1 + \text{Thrust} (d_3 \cos \delta + d_2 \sin \delta)] / F_n\end{aligned}\tag{3}$$

Tipoff begins when the missile CG travels closer than the distance u_1 to the end of the launch tube.

The missile gains a DOF during the tipoff phase of launch. Figure 6 shows the kinematic relationships for this launch phase, and Figure 7 shows the free body forces.




Figure 6. Tipoff Phase Kinematic Relationships

Note that in Figure 6, the origin of the inertial reference frame is initially collinear to the missile CG, a distance d_1 above the bottom of the launch tube. Also note that d_4 , the length of the missile, is also the length of the launch tube. The vector u^2 is the distance from the point of contact between the missile and launch tube and the missile CG. The Y_m component of u^2 corresponds to d_1 , while the X_m component varies with time.




Figure 7. Tipoff Phase Free Body Diagram

The transformation from the missile reference frame to the inertial reference frame is as shown below:

$$\begin{Bmatrix} X \\ Y \end{Bmatrix} = \begin{bmatrix} \cos\theta & -\sin\theta \\ \sin\theta & \cos\theta \end{bmatrix} \begin{Bmatrix} X_m \\ Y_m \end{Bmatrix} \quad (4)$$

The displacement of the missile CG, written in the inertial reference frame, can be expressed as follows:

$$\text{Disp}_{cg} = (d4 + u\hat{2}x\cos\theta - u\hat{2}y\sin\theta)X + (-d1 + u\hat{2}x\sin\theta + u\hat{2}y\cos\theta)Y \quad (5)$$

The velocity of the missile CG is obtained by differentiation of Equation (5) with respect to time, realizing that $u\hat{2}y$ is constant.

$$\begin{aligned} V_{cg} = & (u\hat{2}x\cos\theta - u\hat{2}x\dot{\theta}\sin\theta - \dot{\theta}u\hat{2}y\cos\theta)X \\ & + (u\hat{2}x\sin\theta + u\hat{2}x\dot{\theta}\cos\theta - \dot{\theta}u\hat{2}y\sin\theta)Y \end{aligned} \quad (6)$$

Differentiation of Equation (6), with respect to time, produces an expression for the acceleration of the missile.

$$\begin{aligned}
 a_{cg} = & (\hat{u}\hat{2}x \cos \theta - 2\hat{u}\hat{2}x\dot{\theta} \sin \theta - \hat{u}\hat{2}x\ddot{\theta} \sin \theta \\
 & - \hat{u}\hat{2}x\dot{\theta}^2 \cos \theta - \hat{u}\hat{2}y\ddot{\theta} \cos \theta + \hat{u}\hat{2}y\dot{\theta}^2 \sin \theta)X \\
 & + (\hat{u}\hat{2}x \sin \theta + 2\hat{u}\hat{2}x\dot{\theta} \cos \theta + \hat{u}\hat{2}x\ddot{\theta} \cos \theta \\
 & - \hat{u}\hat{2}x\dot{\theta}^2 \sin \theta - \hat{u}\hat{2}y\ddot{\theta} \sin \theta - \hat{u}\hat{2}y\dot{\theta}^2 \cos \theta)Y
 \end{aligned} \tag{7}$$

Lower order terms in Equation (7) can be combined so that the acceleration relationship can be written more simply.

$$\begin{aligned}
 a_{cg} = & (\hat{u}\hat{2}x \cos \theta + A1\ddot{\theta} + A2)X \\
 & + (\hat{u}\hat{2}x \sin \theta + A3\ddot{\theta} + A4)Y
 \end{aligned}$$

where

$$\begin{aligned}
 A1 &= -\hat{u}\hat{2}y \cos \theta - \hat{u}\hat{2}x \sin \theta, \\
 A2 &= -\dot{\theta}^2 \hat{u}\hat{2}x \cos \theta + \dot{\theta}^2 \hat{u}\hat{2}y \sin \theta - 2 \hat{u}\hat{2}x\dot{\theta} \sin \theta, \\
 A3 &= -\hat{u}\hat{2}y \sin \theta + \hat{u}\hat{2}x \cos \theta, \\
 A4 &= -\hat{u}\hat{2}x\dot{\theta}^2 \sin \theta - \dot{\theta}^2 \hat{u}\hat{2}y \cos \theta + 2\hat{u}\hat{2}x\dot{\theta} \cos \theta.
 \end{aligned} \tag{8}$$

The force and moment summations on the missile, referring to the free body diagram in Figure 7, can be written as follows:

$$\begin{aligned}
 \sum F_x &= \text{Thrust} \cos \theta + FN (-\mu \cos \theta - \sin \theta) - W \sin(QE) = m\ddot{X} \\
 \sum F_y &= \text{Thrust} \sin \theta + FN (-\mu \sin \theta + \cos \theta) - W \cos(QE) = m\ddot{Y} \\
 \sum M_z &= -FNu\hat{2}x - \mu FN d/2 = I_{zz}\ddot{\theta}.
 \end{aligned} \tag{9}$$

Equation (8) is substituted into Equation (9) to produce the complete set of the equations of motion:

$$\begin{aligned}
 FN - (\mu \cos \theta - \sin \theta) - \hat{u}\hat{2}x (m \cos \theta) - mA1\ddot{\theta} &= mA2 - \text{Thrust} \cos \theta + W \sin(QE) \\
 FN - (\mu \sin \theta + \cos \theta) - \hat{u}\hat{2}x (m \sin \theta) - mA3\ddot{\theta} &= mA4 - \text{Thrust} \sin \theta + W \cos(QE) \\
 FN (\hat{u}\hat{2}x + \mu d/2) + I_{zz}\ddot{\theta} &= 0.
 \end{aligned} \tag{10}$$

The equations of motion can be written more simply by collecting the lower order terms as follows:

$$A5 \text{ FN} + A6 \ddot{u}_x + A7 \ddot{\theta} = A8$$

$$A9 \text{ FN} + A10 \ddot{u}_x + A11 \ddot{\theta} = A12$$

$$A13 \text{ FN} + I_{zz} = 0$$

where

$$A5 = -\mu \cos \theta - \sin \theta$$

$$A6 = -m \cos \theta$$

$$A7 = -mA1$$

$$A8 = mA2 + W \sin(QE) - \text{Thrust} \cos \theta$$

$$A9 = -\mu \sin \theta + \cos \theta$$

$$A10 = -m \sin \theta$$

$$A11 = -mA3$$

$$A12 = mA4 + W \cos(QE) - \text{Thrust} \sin \theta$$

$$A13 = \ddot{u}_x + \mu d5/2. \quad (11)$$

Since the equation set describing this system contains only three unknowns, it is not difficult to decouple. This is accomplished by use of Gaussian row reduction, where the equation set is written as follows in matrix form:

$$\begin{matrix} \text{FN} & \ddot{u}_x & \ddot{\theta} \\ \left[\begin{array}{ccc|c} A5 & A6 & A7 & A8 \\ A9 & A10 & A11 & A12 \\ A13 & 0 & I_{zz} & 0 \end{array} \right] \end{matrix} \quad (12)$$

Row 2 of Equation (12) can be replaced with the quantity (Row 1*A9 - Row 2*A5), zeroing the first column of the new Row 2.

$$\left[\begin{array}{ccc|c} A5 & A6 & A7 & A8 \\ 0 & A14 & A15 & A16 \\ A13 & 0 & I_{zz} & 0 \end{array} \right] \quad (13)$$

where $A14 = A9A6 - A5A10$

$$A15 = A9A7 - A5A11$$

$$A16 = A9A8 - A5A12$$

Row 3 of Equation (13) can be replaced with the quantity (Row 1*A13 - Row 3*A5), zeroing the first column of Row 3.

$$\begin{bmatrix} A5 & A6 & A7 & : & A8 \\ 0 & A14 & A15 & : & A16 \\ 0 & A17 & A18 & : & A19 \end{bmatrix} \quad (14)$$

where $A17 = A13A6$

$A18 = A13A7 - A5I_{zz}$

$A19 = A13A8$

Row 3 of Equation (14) can be replaced with the quantity (Row 2*A17 - Row 3*A14) to complete the reduction process.

$$\begin{bmatrix} A5 & A6 & A7 & : & A8 \\ 0 & A14 & A15 & : & A16 \\ 0 & 0 & A17A15 - A14A18 & : & A17A16 - A14A19 \end{bmatrix} \quad (15)$$

The angular acceleration of the missile can now be written directly:

$$\ddot{\theta} = \frac{A17A16 - A14A19}{A17A15 - A14A18}. \quad (16)$$

The linear acceleration of the missile can be expressed in the following fashion:

$$\hat{u}_{2x} = (A16 - A15\ddot{\theta})/A14. \quad (17)$$

The normal force between the missile and launch tube can be written similarly:

$$FN = (A8 - A7\ddot{\theta} - A6 \hat{u}_{2x})/A5. \quad (18)$$

Equations (16) and (17) are integrated numerically to calculate the angular and linear velocities and displacements of the missile as functions of time.

III. SIMULATION RESULTS FOR MISSILE SLUG TEST

The simulation input parameters for the TACAWS missile slug shown in Figure 1 are included in this report in the Appendix. Figure 8 shows three result cases, all simulating launches at a zero QE. The results show the effect of thrust misalignment on missile tipoff rate. In the first case, the thrust misalignment angle is $+0.5$ degree, in the second case the misalignment angle is 0 , and in the third case it is -0.5 degree. Figure 9 shows the effect of thrust misalignment when the launch QE is 10 degree.




Figure 8. Tipoff Rates for TACAWS Slug Missile - Zero QE




Figure 9. Tipoff Rates for TACAWS Slug Missile - 10 Degree QE

Comparison of the results in Figures 8 and 9, shows that tipoff rate is a weak function of QE when QE varies between 0 and 10 degree. However, tipoff rate is highly sensitive to thrust misalignment. Results shown in Figures 8 and 9 indicate that when the thrust misalignment is -0.5 degree, the tipoff rate is -10 degree/second. If the thrust misalignment angle is changed to +0.5 degree, the tipoff rate increases to -13.5 degree/sec. Results for the technology demonstration missile are remarkably similar, given the differences in the missiles in thrust curve, weight, and pitch moment of inertia.

IV. SIMULATION RESULTS FOR TECHNOLOGY DEMONSTRATION MISSILE

The simulation input file for the technology demonstration TACAWS missile is included in the Appendix to this report. Comparison of this input file to the slug test input file reveals that the technology demonstration missile is heavier than the slug missile, has a greater thrust level, and possesses a higher pitch moment of inertia. These differences cancel each other out to some extent, as is shown by Figures 10 and 11. These figures, which show the influence of launch QE and thrust misalignment on tipoff rate, are quite similar to Figures 8 and 9. The influence of thrust misalignment on the technology demonstration missile is less pronounced than on the slug missile.




Figure 10. Tipoff Rate for TACAWS Technology Demonstration Missile - Zero QE




Figure 11. Tipoff Rate for TACAWS Technology Demonstration Missile - 10 Degree QE

As was the case with the slug missile, Figures 10 and 11 show that QE angle does not have a strong influence on tipoff rate over the interval studied. The thrust misalignment angle has a significant influence on tipoff rate, although the greater moment of inertia of the technology demonstration missile reduces its influence.

V. SLUG TEST TIPOFF RESULTS

The TACAWS slug missile was test fired on September 7, 1995. The thrust level of the JAVELIN motors powering the slug was less than was expected. Motor thrust data provided after the test for the as-fired motor configuration indicated that the thrust level was 750 to 800 lbf. High speed film data showing a slug exit velocity of 35 feet/second confirmed the low thrust level. The measured slug tipoff at tube exit was $-0.2 \text{ degree} \pm 0.25 \text{ degree}$. The film resolution was insufficient to estimate tipoff rate. If the simulated motor thrust is decreased to 700 lbf to match the slug exit velocity of 35 feet/second, the simulated tipoff rate is $-18 \text{ degree/second}$ and the simulated tipoff angle is -0.47 degree . This is within the range measured by the high speed film for the actual slug.

VI. CONCLUSIONS

The pitch plane tipoff for the TACAWS technical demonstration missile is in the range -8 degree/second to $-11 \text{ degree/second}$, depending on the angle of thrust misalignment. Gravity has the largest influence on tipoff, contributing $-9.3 \text{ degree/second}$. These results can be seen graphically in Figure 10. The considerable thrust misalignment allowed for TACAWS can influence tipoff significantly, as was shown in Figures 8 through 11. The rigid body dynamic simulation which produced this information was verified by the TACAWS slug test as producing realistic tipoff predictions. The slug test also verified that roll of the slug missile due to fin deployment is too small to be measurable. The simulation was developed under the assumptions of negligible azimuth deflection of the missile, and negligible motion of the launcher during firing. Both of these assumptions were verified as well by the slug test.

APPENDIX
SIMULATION INPUT PARAMETERS FOR THE TACAWS MISSILE SLUG

APPENDIX

SIMULATION INPUT PARAMETERS FOR THE TACAWS MISSILE SLUG

```

program tacaws
C   THIS IS A RIGID BODY SIMULATION OF A TACAWS LAUNCH.  IT MEASURES
C   TIPOFF, AND ASSUMES A RIGID, STATIONARY LAUNCHER.
C   THE MODEL IS TWO-DIMENSIONAL, IN THE PITCH PLANE, AND MODELS
C   THE MISSILE AS A LUMPED MASS.

C   POINT OF CONTACT FOR THE SIMULATION AND ANALYSIS IS
C   DAVID BITTLE
C   AMSMI-RD-ST-SA
C   (205) 876-9410

C   12 DECEMBER, 1994
C   VERSION 1.1

C   VERSION 1.0 MODELED THE MISSILE AS HAVING TWO SHOES.
C   VERSION 1.1 MODELS THE MISSILE AS HAVING NO SHOES AND TIPPING
C   OFF THE END OF THE LAUNCH TUBE.
C   *****

implicit doubleprecision(a-h,o-z)

real*8 lt,izz,nus,nud,m

open(unit=1,file='tacaws.in',status='unknown')
open(unit=2,file='tacaws.out',status='unknown')

1  format(f8.4)
2  format(1x,4(2x,f10.4))

C   Read physical characteristics of missile

read(1,1)
read(1,1)
read(1,1) d1
d1=d1/12.
read(1,1) d2
d2=d2/12.
read(1,1) d3
d3=d3/12.
read(1,1) d4
d4=d4/12.
read(1,1) d5
d5=d5/12.
read(1,1) lt
lt=lt/12.
read(1,1) w
m = w/32.2
read(1,1) det
read(1,1) izz
izz=izz/144.
read(1,1) nus
read(1,1) nud
read(1,1) del
read(1,1) qe
qe=qe/57.3
read(1,1) tlevel
read(1,1) t1

C   ZERO ALL INTEGRATION QUANTITIES PRIOR TO USE

xdd = 0.

```

```

xd = 0.
x = 0.
thdd = 0.
thd = 0.
th = 0.
u2xdd = 0.
u2xd = 0.
u2x = 0.

C   CALCULATE THRUST LEVEL AT DETENT RELEASE

thrust = det+w*(nus*cos(qe)+sin(qe))

if (thrust .le. tlevel) then
  t = thrust*t1/tlevel
else
  write(0,*) 'THRUST NOT SUFFICIENT TO BREAK DETENT'
  stop
endif

C   SIMULATE GUIDANCE PORTION OF LAUNCH

u1 = nud*d5/2.

10  if (x+d2+u1 .lt. lt) then

    if (t .le. t1) then
      thrust = tlevel*t/t1
    else
      thrust = tlevel
    end if
    if (t .gt. .115) thrust = 0.

    xdd1 = (thrust-w*(nud*cos(qe)+sin(qe)))/m
    xd1 = xd+(xdd1+xdd)*del/2.
    x1 = x+(xd1+xd)*del/2.

    xdd = xdd1
    xd = xd1
    x = x1

    write(0,2) t,x*12.,xd
    write(2,2) t,x*12.,xd
    t = t+del
    goto 10
  end if

C   SIMULATE TIPOFF PORTION OF LAUNCH

u2xd = xd
u2x = x+d2-lt
u2y = d5/2.

20  if (u2x .lt. d2) then

    if (t .le. t1) then
      thrust = tlevel*t/t1
    else
      thrust = tlevel
    end if

```

```

      if (t .gt. .115) thrust = 0.

      a1 = -u2y*cos(th)-u2x*sin(th)
      a2 = -thd**2.*u2x*cos(th)+thd**2.*u2y*sin(th)-2.*thd*u2xd*sin(th)
      a3 = -u2y*sin(th)+u2x*cos(th)
      a4 = -thd**2.*u2x*sin(th)-thd**2.*u2y*cos(th)+2.*thd*u2xd*cos(th)
      a5 = -nud*cos(th)-sin(th)
      a6 = -m*cos(th)
      a7 = -m*a1
      a8 = m*a2+w*sin(qe)-thrust*cos(th)
      a9 = -nud*sin(th)+cos(th)
      a10 = -m*sin(th)
      a11 = -m*a3
      a12 = m*a4+w*cos(qe)-thrust*sin(th)
      a13 = u2x+nud*d5/2.
      a14 = a9*a6-a5*a10
      a15 = a9*a7-a5*a11
      a16 = a9*a8-a5*a12
      a17 = a13*a6
      a18 = a13*a7-a5*izz
      a19 = a13*a8

      thdd1 = (a17*a16-a14*a19)/(a17*a15-a18*a14)
      thd1 = thd+(thdd1+thdd)*del/2.
      th1 = th+(thd1+thd)*del/2.

      thdd = thdd1
      thd = thd1
      th = th1

      u2xdd1 = (a16-a15*thdd1)/a14
      u2xd1 = u2xd+(u2xdd1+u2xdd)*del/2.
      u2x1 = u2x+(u2xd1+u2xd)*del/2.

      u2xdd = u2xdd1
      u2xd = u2xd1
      u2x = u2x1

      fn = (a8-a7*thdd1-a6*u2xdd1)/a5

      write(0,2) t,thd1*57.3,u2xd,(u2x-d2)*sin(th)*12.
      write(2,2) t,thd1*57.3,u2xd,(u2x-d2)*sin(th)*12.
      t = t+del
      goto 20
    end if

    stop
  end

```


DATA FILE TACAWS.OUT

TIME (sec)	TIPOFF RATE (deg/sec)	VELOCITY (ft/sec)	AFT RISE (in)
0.1444	-0.0116	26.0175	0.0000
0.1454	-0.0566	26.0077	0.0000
0.1464	-0.1450	25.9976	0.0000
0.1474	-0.2765	25.9874	0.0001
0.1484	-0.4502	25.9771	0.0002
0.1494	-0.6652	25.9667	0.0004
0.1504	-0.9202	25.9562	0.0007
0.1514	-1.2136	25.9456	0.0010
0.1524	-1.5435	25.9350	0.0015
0.1534	-1.9081	25.9245	0.0020
0.1544	-2.3050	25.9139	0.0026
0.1554	-2.7320	25.9035	0.0033
0.1564	-3.1866	25.8931	0.0041
0.1574	-3.6662	25.8828	0.0051
0.1584	-4.1683	25.8726	0.0061
0.1594	-4.6904	25.8626	0.0073
0.1604	-5.2298	25.8528	0.0085
0.1614	-5.7840	25.8431	0.0098
0.1624	-6.3506	25.8337	0.0113
0.1634	-6.9271	25.8245	0.0128
0.1644	-7.5114	25.8154	0.0144
0.1654	-8.1013	25.8066	0.0160
0.1664	-8.6946	25.7981	0.0178
0.1674	-9.2897	25.7898	0.0195
0.1684	-9.8845	25.7817	0.0214
0.1694	-10.4777	25.7739	0.0232
0.1704	-11.0676	25.7663	0.0251
0.1714	-11.6530	25.7590	0.0270
0.1724	-12.2327	25.7520	0.0289
0.1734	-12.8056	25.7452	0.0307
0.1744	-13.3707	25.7386	0.0326
0.1754	-13.9273	25.7323	0.0344
0.1764	-14.4746	25.7262	0.0362
0.1774	-15.0120	25.7204	0.0380
0.1784	-15.5391	25.7148	0.0396
0.1794	-16.0555	25.7094	0.0412
0.1804	-16.5607	25.7043	0.0428
0.1814	-17.0546	25.6994	0.0442
0.1824	-17.5370	25.6946	0.0455
0.1834	-18.0078	25.6901	0.0467
0.1844	-18.4669	25.6858	0.0478
0.1854	-18.9143	25.6817	0.0488
0.1864	-19.3500	25.6778	0.0496
0.1874	-19.7741	25.6740	0.0503
0.1884	-20.1867	25.6705	0.0508
0.1894	-20.5879	25.6671	0.0511
0.1904	-20.9779	25.6638	0.0513
0.1914	-21.3569	25.6608	0.0513
0.1924	-21.7251	25.6579	0.0511
0.1934	-22.0826	25.6551	0.0507
0.1944	-22.4297	25.6525	0.0502
0.1954	-22.7667	25.6500	0.0494
0.1964	-23.0937	25.6477	0.0484
0.1974	-23.4110	25.6455	0.0471
0.1984	-23.7189	25.6434	0.0457
0.1994	-24.0177	25.6415	0.0440

0.2004	-24.3074	25.6396	0.0421
0.2014	-24.5885	25.6379	0.0399
0.2024	-24.8612	25.6363	0.0375
0.2034	-25.1257	25.6348	0.0348
0.2044	-25.3822	25.6334	0.0319
0.2054	-25.6310	25.6321	0.0287
0.2064	-25.8724	25.6310	0.0253
0.2074	-26.1065	25.6299	0.0215
0.2084	-26.3336	25.6289	0.0175
0.2094	-26.5540	25.6280	0.0133
0.2104	-26.7678	25.6271	0.0087
0.2114	-26.9752	25.6264	0.0039
0.2124	-27.1764	25.6258	-0.0013

TACAWS SIMULATION INPUT FILE

8.0	DISTANCE FROM REAR SABOT TO AFT OF MISSILE (IN)
20.52	DISTANCE FROM CG TO AFT OF MISSILE (IN)
23.6	DISTANCE FROM FRONT SABOT TO AFT OF MISSILE(IN)
4 0 <i>→ 4.0</i>	LENGTH OF MISSILE (IN)
48.0	LENGTH OF LAUNCH TUBE (IN)
63.0	WEIGHT OF MISSILE (LBS)
350.0	DETENT SHEAR FORCE (LBS)
311.0	PITCH MOMENT OF INERTIA OF MISSILE ABOUT CG (LB-IN ²)
0.6	STATIC COEFFICIENT OF FRICTION BETWEEN MISSILE AND LAUNCH TUBE
0.3	DYNAMIC COEFFICIENT OF FRICTION BETWEEN MISSILE AND LAUNCH TUBE
0.001	INTEGRATION TIMESTEP FOR SIMULATION

INITIAL DISTRIBUTION LIST

		<u>Copies</u>
IIT Research Institute ATTN: GACIAC 10 W. 35th Street Chicago, IL 60616		1
AMSMI-RD		1
AMSMI-RD-CS-R		15
AMSMI-RD-CS-T		1
AMSMI-RD-MG-TD,	Mr. Mark Dixon	1
	Mr. Walt Jordan	1
AMSMI-RD-ST,	Mr. Doyle	1
AMSMI-RD-ST-SA,	Mr. Bittle	1
	Mr. Cothran	1
	Mr. Nourse	1
AMSMI-GC-IP,	Mr. Fred Bush	1